NAG Library Routine Document

C02AKF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

C02AKF determines the roots of a cubic equation with real coefficients.

2 Specification

```
SUBROUTINE C02AKF(U, R, S, T, ZEROR, ZEROI, ERREST, IFAIL)
INTEGER           IFAIL
double precision  U, R, S, T, ZEROR(3), ZEROI(3), ERREST(3)
```

3 Description

C02AKF attempts to find the roots of the cubic equation

$$uz^3 + rz^2 + sz + t = 0,$$

where u, r, s and t are real coefficients with $u \neq 0$. The roots are located by finding the eigenvalues of the associated 3 by 3 (upper Hessenberg) companion matrix H given by

$$H = \begin{pmatrix} 0 & 0 & -t/u \\ 1 & 0 & -s/u \\ 0 & 1 & -r/u \end{pmatrix}.$$

The eigenvalues are obtained by a call to F08PEF (DHSEQR). Further details can be found in Section 8.

To obtain the roots of a quartic equation, C02ALF can be used.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: U – ***double precision*** *Input*

On entry: u, the coefficient of z^3.

Constraint: U $\neq 0.0$.

2: R – ***double precision*** *Input*

On entry: r, the coefficient of z^2.

3: S – ***double precision*** *Input*

On entry: s, the coefficient of z.

4: T – ***double precision*** *Input*

On entry: t, the constant coefficient.

5: ZEROR(3) – ***double precision*** array *Output*
6: ZEROI(3) – ***double precision*** array *Output*

On exit: ZEROR(i) and ZEROI(i) contain the real and imaginary parts, respectively, of the ith root.

7: ERREST(3) – ***double precision*** array *Output*

On exit: ERREST(i) contains an approximate error estimate for the ith root.

8: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, U $= 0.0$.

IFAIL $= 2$

The companion matrix H cannot be formed without overflow.

IFAIL $= 3$

The iterative procedure used to determine the eigenvalues has failed to converge.

7 Accuracy

If IFAIL $= 0$ on exit, then the ith computed root should have approximately $|\log_{10}(\text{ERREST}(i))|$ correct significant digits.

8 Further Comments

The method used by the routine consists of the following steps, which are performed by routines from LAPACK in Chapter F08.

(a) Form matrix H.

(b) Apply a diagonal similarity transformation to H (to give H').

(c) Calculate the eigenvalues and Schur factorization of H'.

(d) Calculate the left and right eigenvectors of H'.

(e) Estimate reciprocal condition numbers for all the eigenvalues of H'.

(f) Calculate approximate error estimates for all the eigenvalues of H' (using the 1-norm).

9 Example

This example finds the roots of the cubic equation

$$z^3 + 3z^2 + 9z - 13 = 0.$$

9.1 Program Text

```
*     C02AKF Example Program Text
*     Mark 20 Release. NAG Copyright 2001.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      DOUBLE PRECISION R, S, T, U
      INTEGER          I, IFAIL
*     .. Local Arrays ..
      DOUBLE PRECISION ERREST(3), ZEROI(3), ZEROR(3)
*     .. External Subroutines ..
      EXTERNAL         C02AKF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'C02AKF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) U, R, S, T
      IFAIL = 1
*
      CALL C02AKF(U,R,S,T,ZEROR,ZEROI,ERREST,IFAIL)
*
      IF (IFAIL.EQ.0) THEN
         WRITE (NOUT,*)
         WRITE (NOUT,*) ' Roots of cubic equation     ',
     +     '        Error estimates'
         WRITE (NOUT,*) '                             ',
     +     '      (machine-dependent)'
         WRITE (NOUT,*)
         DO 20 I = 1, 3
            WRITE (NOUT,99999) ' z = ', ZEROR(I), ZEROI(I), '*i',
     +        ERREST(I)
   20    CONTINUE
      ELSE
         WRITE (NOUT,*)
         WRITE (NOUT,99998) ' ** C02AKF returned with IFAIL = ', IFAIL
      END IF
*
99999 FORMAT (1X,A,1P,E12.4,SP,E12.4,A,8X,SS,E9.1)
99998 FORMAT (1X,A,I5)
      END
```

9.2 Program Data

```
C02AKF Example Program Data
 1.0   3.0   9.0  -13.0 : Values of U, R, S and T
```

9.3 Program Results

```
 C02AKF Example Program Results

  Roots of cubic equation            Error estimates
                                   (machine-dependent)

  z =   1.0000E+00 +0.0000E+00*i          1.0E-15
  z =  -2.0000E+00 +3.0000E+00*i          1.0E-15
  z =  -2.0000E+00 -3.0000E+00*i          1.0E-15
```